
To:  File   
    
From: John Effland   
    
Date: 2005-03-10   
 
Revisions: 2005-03-10 jee Initial 
 2006-02-24 jee SW Version 1.1 now includes the leave type “Other” 
 
Subject: Leave Calendar Program Architecture and Design 
 

 

1. Summary 

A program to generate a calendar that displays vacation and travel times for employees was written in August of 
2000.  It was subsequently updated to show the schedule of visitors to the NTC.  The leave calendar software 
presents the user with a dialog box to assign leave as shown in Figure 1, stores the leave dates and names in a 
database, and then queries the database to generate static HTML pages which display calendars for the current and 
following month.  A sample calendar generated by this program is shown in Figure 2.  While access to the database 
is restricted to a few users, anyone can access the HTML calendar files residing on \\cvfiler\cv-cdl-pub. This 
document describes the design of the software and the system architecture. 
 
It is important to note that the architecture is basically obsolete and should eventually be replaced by a 
commercially available, dynamic web-based design. 

2. History 

The software originated from an article in Microsoft Developer’s Net News Magazine1 entitled “Building a Web-
based Event Calendar.”  That article assumed the developer would generate Active Server Pages on Microsoft’s 
Internet Information Server, so much of the code was written for that platform.  The NRAO didn’t have that web 
server, so the code was ported to Visual Basic Version 6 to build the executable.  A SQL database server was 
running on my office computer “wheat” and was used to store the data and stored procedures, which also reside in 
the database server.   
 
The intent of the MSDN article was to build a dynamic, web-based scheduling program, but this was scaled down 
to a simple executable program running on a windows-based client computer, accessing a server-based database, 
and generating static HTML code to a file each time the program is run. 

3. Architecture 

The overall architecture of the program is shown in Figure 3.  The executable file is stored on \\cvfiler and is 
downloaded and run on the client computer.  The user must have access rights to the SQL server program running 
on wheat.  Data are returned from this database server to the client in the form of “recordsets” using Microsoft’s 

                                                      
1 MSDN News Magazine, July/August 2000, Volume 9, Number 4, Page 7. 

API called “Active-X Data Object’s” (ADO).  The static HTML calendar page is built and the resulting file is 
stored on \\cvfiler\cv-cdl-pub with a file name that includes the year and month, such as Leave2005Apr.html. 
 
The dialog box shown in Figure 1 uses several Visual Basic controls that require further discussion.  A hidden 
ADO data control allows inserting records into the database and also simplifies the connection of the datagrid to the 
database.  Figure 4 shows the form as seen from the Visual Basic IDE and the data control, named adodc1, is 
hidden when the program runs.  The data control updates the database with user names and leave dates in routine 
btnAddLeave_Click.  Existing names and leave dates are tabulated in the datagrid on the dialog box, which allows 
the user to edit data directly in the cells of the grid.  Changes the user makes to entries in the datagrid are posted 
back into the database via linkage by defining the data control to be the datasource for the datagrid. 
 
The data control is dynamically loaded with the dataset connection string in routine Form_Load and its recordsource 
is defined in the same routine.  Attempts to dynamically generate the recordsource by returning a string from a 
stored procedure were never successful, but the code remains commented in the routine for future reference. 
 
The leave dates are selected using a DateTimePicker control. 
 
The stored procedures, listed in Section 7, are routines written in Microsoft’s “T-SQL,” run on the database server, 
and return recordsets.  The Visual Basic code is listed in Section 8. 

4. File Locations 

The leave calendar software uses a number of files given in this section. 
 
Visual Basic VB Project files: 
 
\\cvfiler.cv.nrao.edu\cv-cdl-sis\MeasSys\Software\Utils\WebCalendar\WebCalCode.vbp 
\\cvfiler.cv.nrao.edu\cv-cdl-sis\MeasSys\Software\Utils\WebCalendar\WebCalCode.vbw 
 
Visual Basic form file: 
 
\\cvfiler.cv.nrao.edu\cv-cdl-sis\MeasSys\Software\Utils\WebCalendar\Form1.frm 
 
Visual Basic executable location: 
 
\\cvfiler.cv.nrao.edu \cv-cdl-pub\WebCalendarSoftware\WebCalendar.exe 
 
 
Directory for HTML Leave Calendars 
 
\\cvfiler.cv.nrao.edu\cv-cdl-pub\ 

5. Database Use 

Names and dates are stored on a MS SQL Server version 7.02 database running on computer wheat in my office. 
 
Database name:  TeamWeb 
Tables Used:  Schedule 
 
The database table Schedule is structured as shown in Table 1. 
                                                      
2 Wheat is actually running Microsoft’s Data Engine, which is functionally equivalent to SQL Server but the number of 
concurrent users is limited by slowing down the server when more than 5 concurrent users are logged into the server. 

Table 1: Schema for Database Table “Schedule” 

Field Name Datatype Size (Bytes) 
idSchedule smallint 2 
dtDateStart smalldatetime 4 
vcEvent varchar(100) 100 
dtDateEnd smalldatetime 4 
iEventType smallint 2 

 
A number of other tables in the database TeamWeb were for prototyping of other projects but are now unused. 

6. Windows Libraries and OLE Objects 

The leave calendar software uses the following libraries and OLE objects: 
 
stdole2.tlb  OLE Automation 
MSBIND.DLL  Microsoft Data Binding Collection VB 6.0 (SP4) 
msador15.dll  Microsoft ActiveX Data Objects Recordset 2.0 Library 
msado25.tlb  Microsoft ActiveX Data Objects 2.0 Library 
MSADODC.OCX 
MSCOMCTL.OCX 
MSCOMCT2.OCX 
MSDATGRD.OCX  

Figure 1: Dialog Box for adding leave to the database and generating the calendar file 

 

 
Figure 2: Sample of Calendar Output 

Figure 3: Overall Architecture 

Figure 4: Form from Visual Basic Editor 

7. Stored Procedures in Database “TeamWeb” on MS SQL Server “Wheat” 

The following stored procedures run queries on tables in the database TeamWeb and return recordsets using 
Microsoft’s ADO API.  Note that only the procedure spGetScheduleforDay actually runs with the leave calendar 
software.  The other stored procedures are uncompleted enhancements. 
 
***************************************************************** 
CREATE Procedure spGetScheduleForDay(@nDateTime smalldatetime) 
-- Returns all events whose dates fall between the date given in the parameter 
-- 2000-08-08 jee 
As 
 SELECT vcEvent, iEventType 
 FROM Schedule 
 -- Style 0 shows dates as text in form: "mon dd yyyy hh:miAM (or PM)" 
 WHERE (CONVERT(DATETIME, @nDateTime, 0)  
     BETWEEN CONVERT(DATETIME, dtDateStart,0) AND CONVERT(DATETIME, dtDateEnd,0)) 
 ORDER BY vcEvent 
Return 
 
**************************************************************** 
CREATE Procedure spGetRecsByDate 
-- Returns all events ordered by date 
-- 2000-09-01 jee 
As 
 SELECT * 
 FROM Schedule 
 ORDER BY dtDateStart DESC 
return 
 
**************************************************************** 
CREATE Procedure spGetSchedule(@nMonth tinyint,@nYear smallint) 
 
As 
 select idSchedule, convert(varchar, datepart(dd, dtDateStart)) 'nDay', vcEvent 
 from Schedule 
 where datepart(yy, dtDateStart) = @nYear and datepart(mm, dtDateStart) = @nMonth 
 order by datepart(dd, dtDateStart) 
return 

8. Visual Basic Code 

The following code is used to generate the dialog box to assign vacations and to generate the static web pages. 
 
Option Explicit 
' 
Private Const sVersion = "1.1 2006-02-24" 
' 
' V 1.1    2006-02-24 jee 1) added "Other Leave" type and color 
'                         2) Updated VB to SP6 to get progress bar working again 
' V 1.0.14 2005-03-10 jee returned fonts to orig size and sat/sun columns only 50 wide 
' V 1.0.13 2005-03-09 jee made fonts larger and sat/sun cols smaller. 
' V 1.0.12 2004-08-17 jee changed header from CDL to NTC in DoHeader 
' V 1.0.11 2004-01-14 jee changed header from CDL to NTC in DoHeader 
' V 1.0.10 2003-12-15 jee changed header from CDL to NTC 
' V 1.0.9 2003-09-29 jee changed from Appointments to Visitors in DoFooter and iLeave_Type 
' V 1.0.8 2003-01-22 jee corrected spelling of Shure to sure!! 
' V 1.0.7 2002-11-20 jee cvfiler root is no longer "shares" but is "cv-cdl-pub" 
' V 1.4 2002-10-18 jee changed to use CV Filer,  spGetScheduleForDay now 
'                      sorts by "event" 
' V 1.3 2000-09-01 jee Orders data control records by date 

' V 1.2 2000-08-31 jee corrected color in MakeCalendar color assignments. 
' V 1.1 2000-08-18 jee added error checks for data control. 
' V 1.0 2000-08-16 jee 1) added multiple month support. 
' 
 
 
 
Private Enum iLEAVE_TYPE 
    Vacation = 1 
    Travel 
    Visitor 
    Other 
End Enum 
 
Private Const sCOLOR_VACATION As String = "red" 
Private Const sCOLOR_TRAVEL As String = "black" 
Private Const sCOLOR_VISITOR As String = "green" 
Private Const sCOLOR_OTHER_LEAVE As String = "blue" 
 
' file name for output of html file 
'Private Const sDIRECTORY_NAME As String = "//cvfiler/shares/cv-cdl-pub/" 
Private Const sDIRECTORY_NAME As String = "//cvfiler/cv-cdl-pub/" 
 
' Database connection string 
Private Const sdbCONNECTION = "Provider=SQLOLEDB.1;Integrated Security=SSPI;" & _ 
                              "Persist Security Info=False;Initial Catalog=TeamWeb;Data 
Source=wheat" 
                               
Private m_oConnection As ADODB.Connection 
                               
' if TRUE, don't change the end date to match changes in the start date. 
Private bEndDateChanged As Boolean 
 
' if true, quit 
Private m_bDataBaseError As Boolean 
 
'Private m_rsEvents As ADODB.Recordset 
 
'******************************************************************** 
'******************************************************************** 
Private Sub Adodc1_Error(ByVal ErrorNumber As Long, Description As String, ByVal Scode As 
Long, ByVal Source As String, ByVal HelpFile As String, ByVal HelpContext As Long, 
fCancelDisplay As Boolean) 
    ' Runs if error occurs with Active X Data Object Control 
    ' 
    ' 2000-08-18 jee 
    ' 
    Call MsgBox(Description, vbInformation + vbOKOnly) 
     
    ' don't display error in another dialog box 
    fCancelDisplay = True 
     
    m_bDataBaseError = True 
End Sub 
'******************************************************************** 
Private Sub btnAddLeave_Click() 
    ' 2000-08-10 jee 
    ' 2000-08-15 jee Moved MakeCalendar to own button 
    ' 2003-09-29 jee changed from Appointments to Visitors 
    ' 2006-02-24 jee 1) added other leave type 
    '                2) now hide status box 
    ' 
    Dim iLeaveTypeBuff As Integer 

    Dim vDateDiff As Variant 
     
    Call ShowStatus("") 
     
    ' validate the field values 
    If Len(Me.tbName) = 0 Then 
        Call MsgBox("Please enter someone's name for this leave.", _ 
                vbOKOnly + vbExclamation) 
        Exit Sub 
    End If 
     
    If Len(Me.DTPickerStart.Value) = 0 Then 
        Call MsgBox("Please enter a valid start date for this leave.", _ 
                vbOKOnly + vbExclamation) 
        Exit Sub 
    End If 
     
    If Len(Me.DTPickerEnd.Value) = 0 Then 
        Call MsgBox("Please enter a valid end date for this leave.", _ 
                vbOKOnly + vbExclamation) 
        Exit Sub 
    End If 
     
    ' compare number of seconds to make sure dates OK 
    vDateDiff = DateDiff("s", Me.DTPickerStart.Value, Me.DTPickerEnd.Value) 
    If DateDiff("s", Me.DTPickerStart.Value, Me.DTPickerEnd.Value) < 0 Then 
        Call MsgBox("The start date occurs after the end date.", _ 
                vbOKOnly + vbExclamation) 
        Exit Sub 
    End If 
     
    ' decode the leave type 
    If Me.obTravel.Value Then 
        iLeaveTypeBuff = iLEAVE_TYPE.Travel 
    ElseIf Me.obVisitor.Value Then 
        iLeaveTypeBuff = iLEAVE_TYPE.Visitor 
    ElseIf Me.obOther.Value Then 
        iLeaveTypeBuff = iLEAVE_TYPE.Other 
    Else 
        iLeaveTypeBuff = iLEAVE_TYPE.Vacation 
    End If 
     
    With Me.Adodc1.Recordset 
        .AddNew 
        !Name = Me.tbName.Text 
        !DateStart = Me.DTPickerStart.Value 
        !DateEnd = Me.DTPickerEnd.Value 
        !iEventType = iLeaveTypeBuff 
        .Update 
    End With 
     
     
    ' reset the input fields 
    Me.tbName.Text = "" 
    Me.DTPickerEnd.Value = Format(Now, "yyyy-MM-dd") 
    Me.DTPickerStart.Value = Format(Now, "yyyy-MM-dd") 
    Me.obVacation.Value = True 
    bEndDateChanged = False 
End Sub 
'******************************************************************** 
Private Sub btnDeleteEntry_Click() 
    ' deletes the selection on the data grid 
    ' 

    ' 2000-08-10 jee 
    ' 2003-01-22 jee corrected shure to sure!! 
    ' 2006-02-24 jee added hide statusbox 
    ' 
    Call ShowStatus("") 
     
    If MsgBox("Are you sure you want to delete the record" _ 
              & vbCr & " highlighted in the grid box?", vbQuestion + vbYesNo) = vbYes Then 
        With Me.Adodc1.Recordset 
            .Delete 
        End With 
    End If 
End Sub 
'******************************************************************** 
Private Sub btnQuit_Click() 
    ' 2000-08-10 jee 
    ' 
    Unload Me 
End Sub 
'******************************************************************** 
Private Sub btUpdateCalendar_Click() 
    ' 2000-08-15 jee 
    ' 2006-02-24 jee Hide status area at first 
    ' 
    Dim sFileName1 As String 
    Dim sFileName2 As String 
    Const sINTERVAL_TYPE As String = "m"    ' "m" specifies months as interval. 
    Dim vThisMonth As Variant   ' current date 
    Dim vNextMonth As Variant   ' a date in the next month 
     
    ' hide the status area 
    Call ShowStatus("") 
     
    vThisMonth = Now 
     
    ' add one month the the current date to get next month's date 
    vNextMonth = DateAdd(sINTERVAL_TYPE, 1, vThisMonth) 
    ' 
    ' Generate file names from date 
    ' first file name is this month's schedule 
    sFileName1 = sDIRECTORY_NAME & "Leave" & Format$(vThisMonth, "YYYYMMM") & ".html" 
    ' second file name is the next months schedule 
    ' This is obtained by adding one month to the current date 
    sFileName2 = sDIRECTORY_NAME & "Leave" & Format$(vNextMonth, "YYYYMMM") & ".html" 
     
    ' Fetch data from database and write out file 
     
    'first, for this month 
    Open sFileName1 For Output As #1 
    Me.ProgressBar.Visible = True 
    Call MakeCalendar(vThisMonth, 0, sFileName2) 
    Me.ProgressBar.Visible = False 
    Close #1 
     
    ' then, for next month 
    Open sFileName2 For Output As #1 
    Me.ProgressBar.Visible = True 
    Call MakeCalendar(vNextMonth, 1, sFileName1) 
    Me.ProgressBar.Visible = False 
    Close #1 
     
    ' update the statusbar 
    Call ShowStatus("Files written to:" & vbCr & vbCr & _ 

                           sFileName1 & vbCr & vbCr & _ 
                           sFileName2) 
 
End Sub 
'******************************************************************** 
Private Sub DTPickerEnd_Change() 
    ' When the end date changes, 
    ' match this with changes in the start date. 
    ' 
    ' 2000-08-10 jee 
    ' 
    ' if TRUE, don't change the end date to match changes in the start date. 
    bEndDateChanged = True 
 
End Sub 
'******************************************************************** 
Private Sub DTPickerStart_Change() 
    ' Change the end date to match this, unless the user 
    ' has already changed the end date. 
    ' 
    ' 2000-08-10 jee 
    ' 
    ' 
    If Not bEndDateChanged Then 
        Me.DTPickerEnd.Value = Me.DTPickerStart.Value 
    End If 
End Sub 
'******************************************************************** 
Private Sub Form_Initialize() 
    ' 
    ' 2000-08-10 jee 
    ' 2000-08-18 JEE added init of data control. 
    ' 
    ' if TRUE, don't change the end date to match changes in the start date. 
    bEndDateChanged = False 
     
    m_bDataBaseError = False 
     
End Sub 
'******************************************************************** 
Private Sub Form_Load() 
    ' 2000-08-10 jee 
    ' 2000-08-18 jee added init of data control 
    ' 2000-09-01 added m_oConnection 
    ' 2003-12-15 jee changed caption from CDL ... to NTC ... 
    ' 
    Me.ProgressBar.Visible = False 
    Me.ProgressBar.Scrolling = ccScrollingSmooth 
     
'    Dim oConn As ADODB.Connection 
     
     
'    Set m_rsEvents = m_oConnection.Execute("spGetRecsByDate") 
     
    ' setup default values 
    Me.DTPickerStart.Value = Format(Now, "yyyy-MM-dd") 
    Me.DTPickerEnd.Value = Format(Now, "yyyy-MM-dd") 
    Me.tbName.Text = "" 
     
    Me.obVacation.Value = True 
    Me.Caption = "NTC Leave Schedule Version " & sVersion 
     
    Me.Show 

    ' load the data control with the connection string 
    Call ShowStatus(vbCr & vbCr & "     Connecting to database...") 
     
    Me.Refresh 
     
'    Set oConn = cnGetDataConnection 
     
    With Me.Adodc1 
        .ConnectionString = sdbCONNECTION 
        ' Can't get a stored procedure to populate the data grid because it produces forward 
only cursors. 
        'Set .Recordset = oConn.Execute("spGetRecsByDate") 
        .RecordSource = "SELECT dtDateStart AS DateStart, dtDateEnd AS DateEnd, vcEvent AS 
Name, iEventType FROM Schedule ORDER BY dtDateStart DESC" 
        On Error Resume Next 
        .Refresh 
        On Error GoTo 0 
    End With 
     
    If m_bDataBaseError Then 
        End 
    End If 
     
    With Me.DataGrid1 
        Set .DataSource = Me.Adodc1 
    End With 
     
   Call ShowStatus("") 
     
     
End Sub 
'******************************************************************** 
Sub ShowStatus(ByVal sStatus) 
    ' shows the status box 
    ' 
    ' Calling Parms: sStatus - string to show 
    '                          if "", then hide status box 
    ' 
    ' 2000-08-18 jee 
    ' 2006-02-24 jee added refresh 
    ' 
    Me.labStatus.Caption = sStatus 
     
    If Len(sStatus) = 0 Then 
        Me.labStatus.Visible = False 
    Else 
        Me.labStatus.BackColor = vbBlue 
        Me.labStatus.ForeColor = vbWhite 
        Me.labStatus.Visible = True 
    End If 
    Me.Refresh 
End Sub 
'******************************************************************** 
Sub DoHeader(ByVal sTitle As String) 
    ' 2004-01-14 jee changed header from CDL to NTC 
    ' 2004-08-17 jee now I really think I changed the header from CDL to NTC 
    ' 
     
    Print #1, "<html>" 
    Print #1, "<head>" 
    Print #1, "<META HTTP-EQUIV=""Content-Type"" CONTENT=""text/html; charset=iso8859-1"">" 
    Print #1, "<title>NTC Leave Calendar -  " & sTitle & "</title>" 

    Print #1, "</head>" 
 
    Print #1, "<body bgcolor=""white"" link=""blue"" alink=""blue"" vlink=""blue"">" 
    Print #1, "<basefont face=""Verdana, Arial"">" 
 
    Print #1, "<center><h1>NTC Leave Calendar</h1>" 
    Print #1, "<h3> " & sTitle & " </h3>"; 
End Sub 
'******************************************************************** 
Sub DoFooter() 
    ' 2000-08-15 jee 
    ' 2000-08-31 jee added software version 
    ' 2003-09-29 jee changed from Appointments to Visitors 
    ' 2006-02-24 jee added general leave color 
    ' 
    Print #1, "</center>" 
    Print #1, "<p> <font size = X small>" & "Color Key:" 
    Print #1, "<UL> <font color=""" & sCOLOR_VACATION & """><LI> " & "Vacation" 
    Print #1, "<font color=""" & sCOLOR_OTHER_LEAVE & """><LI> " & "Other Leave (Family, Jury, 
Doctor, etc.)" 
    Print #1, "<font color=""" & sCOLOR_TRAVEL & """><LI> " & "Travel" 
    Print #1, "<font color=""" & sCOLOR_VISITOR & """><LI> " & "Visitors" & " </UL>" 
    Print #1, "<P> <font color=""black""> " & "Last Modified: " & Format$(Now(), "yyyy-mm-dd 
hh:mm:ss") 
    Print #1, " &nbsp;&nbsp;&nbsp;&nbsp; Software Ver. " & sVersion; "</p>" 
    Print #1, "</body></html>" 
End Sub 
'******************************************************************** 
Function cnGetDataConnection() As ADODB.Connection 
    ' 
    ' 2000-08-07 jee 
    ' 2000-08-10 jee changed connection string to be constant 
    ' 2000-09-01 jee changed to use m_oConnection 
    ' 
'    Dim oConn As ADODB.Connection 
'    Dim strConn As String 
'    Set oConn = CreateObject("ADODB.Connection") 
    If m_oConnection Is Nothing Then 
        Set m_oConnection = CreateObject("ADODB.Connection") 
        m_oConnection.Open sdbCONNECTION 
    End If 
    Set cnGetDataConnection = m_oConnection 
End Function 
'******************************************************************** 
Private Sub MakeCalendar(ByVal dtDate As Variant, _ 
                         ByVal iWhichMonth As Integer, _ 
                         ByVal sLinkFileName As String) 
    ' 
    ' Purpose: Main calendar page that shows events in a calendar format 
    ' 
    ' Calling Parms: vDate - date used to obtain month to query and generate 
    '                        calendar data. 
    '                iWhichMonth = not 1 is current month 
    '                              1 - next month 
    '                sLinkFileName - name of file containing prevous/next month calendar 
    ' 
    ' 2000-08-07 jee from MSDN paper, August, 2000 
    ' 2000-08-10 jee 
    ' 2000-08-16 jee 
    ' 2000-08-31 jee corrected color assignments 
    ' 2005-03-09 jee shortened sat and sun 
    ' 2005-03-10 jee cleaned up do loop 
    ' 2006-02-24 jee added other leave, removed progress bar 

    Dim dbConn, rs, nDex, nMonth, nYear 
     
    ' Set the Month and Year 
    If nMonth = "" Then nMonth = Month(dtDate) 
    If nYear = "" Then nYear = Year(dtDate) 
     
    ' Set the date to the first of the current month 
    dtDate = DateSerial(nYear, nMonth, 1) 
     
    Call ShowStatus("Connecting to database...") 
    Set dbConn = cnGetDataConnection 
    Call ShowStatus("") 
    'Set rs = dbConn.Execute("spGetSchedule " & nMonth & ", " & nYear) 
     
    DoHeader (MonthName(Month(dtDate)) & "&nbsp;&nbsp;" & nYear) 
 
    ' print out the days of the week in the table header 
    Print #1, "<tr><td colspan=2>" 
    Print #1, "<table border=1 bgcolor=""gray"" cellpadding=3>" 
    Print #1, "<tr bgcolor=""Blue"">" 
    Print #1, "<td width=50><font color=""white""><b>Sun</b></font></td>" 
    Print #1, "<td width=180><font color=""white""><b>Monday</b></font></td>" 
    Print #1, "<td width=180><font color=""white""><b>Tuesday</b></font></td>" 
    Print #1, "<td width=180><font color=""white""><b>Wednesday</b></font></td>" 
    Print #1, "<td width=180><font color=""white""><b>Thursday</b></font></td>" 
    Print #1, "<td width=180><font color=""white""><b>Friday</b></font></td>" 
    Print #1, "<td width=50><font color=""white""><b>Sat</b></font></td></tr>" 
    Print #1, "<tr bgcolor=""#ffffc0"">" 
     
    ' Add blank cells until the first day of the month 
    For nDex = 1 To Weekday(dtDate) - 1 
        Print #1, "<td bgcolor=""#c0c0c0"">&nbsp;</td>" 
    Next 
     
    ' Loop for this month 
    With Me.ProgressBar 
        .Min = 0 
        .Max = 31 
        .Visible = True 
        .Value = 0 
    End With 
    Do 
        Me.ProgressBar.Value = Me.ProgressBar.Value + 1 
       ' print out the day of the month at the top of the cell: 
        Print #1, "<td valign=""top"">" & Day(dtDate) & "<br>&nbsp;<br>" 
         
        ' find all entries for this date and put them into the recordset 
         Set rs = dbConn.Execute("spGetScheduleForDay '" & dtDate & "'") 
         
        ' loop until the recordset is empty 
        'If Not rs.EOF Then 
            ' loop until EOF 
            Do Until rs.EOF 
                ' font size is one size smaller 
                Print #1, "<font size=""-1"">" 
                 
                ' change colors according to type 
                Select Case rs.Fields("iEventType") 
                    Case iLEAVE_TYPE.Vacation: 
                        Print #1, "<font color=""" & sCOLOR_VACATION & """>"; 
                    Case iLEAVE_TYPE.Other: 
                        Print #1, "<font color=""" & sCOLOR_OTHER_LEAVE & """>"; 

                    Case iLEAVE_TYPE.Travel: 
                        Print #1, "<font color=""" & sCOLOR_TRAVEL & """>"; 
                    Case iLEAVE_TYPE.Visitor: 
                        Print #1, "<font color=""" & sCOLOR_VISITOR & """>"; 
                    Case Else: 
                        Print #1, "<font color=""black"">"; 
                End Select 
                ' print out the event 
                Print #1, "<b>" & rs("vcEvent"); 
                Print #1, "</b></font><br>" 
                rs.MoveNext 
                 
                'If rs.EOF Then Exit Do 
            Loop 
        'End If 
 
        Print #1, "</td>" 
                 
        ' terminate row if day is Saturday 
        If Weekday(dtDate) = 7 Then 
            Print #1, "</tr>" & vbCrLf & "<tr  bgcolor=""#ffffc0"">" 
        End If 
         
        dtDate = DateAdd("d", 1, dtDate) 
    Loop Until (Month(dtDate) <> CInt(nMonth)) 
 
    ' Add blank cells to fill out the rest of the month if needed 
    If Weekday(dtDate) <> 1 Then 
        For nDex = Weekday(dtDate) To 7 
            Print #1, "<td bgcolor=""#C0C0C0"">&nbsp;</td>" 
        Next 
    End If 
    Print #1, "</tr>" 
    Print #1, "</table></td></tr>" 
     
    Print #1, "<tr><td><a href=""" & sLinkFileName & """ > " 
         
    ' enter previous/next month links 
    If iWhichMonth = 1 Then 
        Print #1, "<b>&lt - Previous Month" 
    Else 
        ' current month 
        Print #1, "<b>Next Month - &gt" 
    End If 
     
    Print #1, "</b></a></td></tr>" 
    Print #1, "</form>" 
     
     
    Call DoFooter 
 
End Sub 


